

A Minimum Wage in a Two-Sector Economy in the Short and Long Run

Jake Cunnane & Yinebeb Girma
5/3/2014

Table of Contents

Introduction...	2
Parameters & Variables...	3
Structure of the Model...	4
Base Case...	5
Analysis of Shocks...	6-10
Appendix A: Our Short Run Model...	11-13
Appendix B: Gilbert and Tower's Long Run Model...	14-16

Introduction

For our project, we have developed a short run version of the small country Harris-Todaro model presented in Chapter 21 of Gilbert and Tower's *Introduction to Numerical Simulation for Trade Theory and Policy*. The Harris-Todaro model simulates an economy in which two goods are produced, one in the city and one in the country. The good produced in the city, henceforth called good 1, is capital-intensive; that produced in the country (good 2) is labor-intensive. A minimum wage is then imposed in the city, while labor in the country and capital in both environments remain unregulated. In the long run version of the model, there is only one kind of capital, K . Like labor, it is mobile between the two sectors. In our short run version, however, capital is *not* mobile—only labor. The capital allotment of each sector is provided exogenously and is fixed. We call urban capital M and rural capital N . By treating the capital stock as two separate factors (M & N), we allow the rental rate of capital in the city and that in the country to diverge in the short run, though in the long run, as we will see, capital will move between the two sectors until the rental rate is the same everywhere. Because we are dealing with a labor market distortion, the model is also set up to allow for unemployment. In our analysis of the short run effects of a minimum wage on a two-sector economy, we will vary several parameters, including the elasticity of substitution between factors, the price of the capital-intensive good, and the level of the imposed minimum wage.

Parameters & Variables

Before we continue, it is necessary to introduce the parameters we will be manipulating and referring to throughout the rest of this paper. The parameters defined are the same in the long and short run versions of the model. In the notation, “I” refers to goods (1 and 2) and “J” refers to factors (K for the long run model and M & N for the short run model).

Parameters	
ALPHA	Shift parameters in utility
BETA(I)	Share parameters in utility
P(I)	Prices
UO	Initial utility level
CO(I)	Initial consumption levels
XO(I)	Initial trade levels
GAMMA(I)	Shift parameters in production
DELTA(J,I)	Share parameters in production
RHO(I)	Elasticity parameters in production
FBAR(J)	Endowments
QO(I)	Initial output levels
RO(J)	Initial factor prices
FO(J,I)	Initial factor use levels
GDPO	Initial gross domestic product
RIO(J,I)	Initial industry level factor prices
PIO(J)	Initial employment rate

The models also use the same endogenous variables, given below:

Variables	
U	Utility index
X(I)	Trade levels
C(I)	Consumption levels
Q(I)	Output levels
R(J)	Factor prices
F(J,I)	Factor use levels
GDP	Gross domestic product
RI(J,I)	Industry factor prices
PI(J)	Employment rate

Structure of the Model

The generic forms of the equations representing the short run system are as follows. For the specific forms, which are taken from the Harris-Todaro long run Harris-Todaro model and the sector-specific capital model in Gilbert and Tower, see Appendix A for the model itself.

1. $c_1(\bar{w}_1, r) = p_1$
2. $c_2(w_2, r) = p_2$
3. $w_2 = \pi \bar{w}_1$
4. $a_{K_1}(\bar{w}_1, r)q_1 = \bar{K}_1$
5. $a_{K_2}(w_2, r)q_2 = \bar{K}_2$
6. $a_{L_1}(\bar{w}_1, r)q_1 + \pi a_{L_2}(w_2, r)q_2 = \pi \bar{L}$

The key component in addition to a binding minimum wage in the urban sector is the introduction of π in the equation 3. This is equivalent to saying that the rural wage is equal to the expected urban wage.

The generic forms of the equations representing the long run system are as follows. For their specific forms, see Appendix B for the model.

1. $c_1(\bar{w}_1, r) = p_1$
2. $c_2(w_2, r) = p_2$
3. $w_2 = \pi \bar{w}_1$
4. $a_{K_1}(\bar{w}_1, r)q_1 + a_{K_2}(w_2, r)q_2 = \bar{K}$
5. $a_{L_1}(\bar{w}_1, r)q_1 + \pi a_{L_2}(w_2, r)q_2 = \pi \bar{L}$

Base Case

Both the long run Harris-Todaro model presented in Chapter 21 of Gilbert and Tower and the short run version we have constructed are set up so that when the imposed minimum wage equals 1, it is only just binding and does not create a distortion. Therefore, when the minimum wage in the city equals 1, the economy remains at full employment. The parameters for our base case are set as follows:

Short Run	Long Run
$P(i)=1;$	$P(i)=1;$
$RO(j)=1;$	$RO(J)=1;$
$QO(I)=100;$	$QO(I)=100;$
$CO(I)=QO(I);$	$CO(I)=QO(I);$
$XO(I)=QO(I)-CO(I);$	$XO(I)=QO(I)-CO(I);$
$FO('L','1')=20;$	$FO('L','1')=20;$
$FO('L','2')=80;$	$FO('L','2')=80;$
$FO('m','1')=80;$	$FO('k','1')=80;$
$FO('n','2')=20;$	$FO('k','2')=20;$
$FBAR(J)=SUM(I, FO(J,I));$	$FBAR(J)=SUM(I, FO(J,I));$
$GDPO=SUM(I, P(I)*QO(I));$	$GDPO=SUM(I, P(I)*QO(I));$
$RHO(I)=.1;$	$RHO(I)=.1;$
$DELTA(J,I) = FO(J,I) * (RO(J)/FO(J,I) ** (RHO(I)-1)) / (SUM(J, FO(J,I) * (RO(J)/FO(J,I) ** (RHO(I)-1))));$	$DELTA(J,I) = (RO(J)/FO(J,I) ** (RHO(I)-1)) / (SUM(J, RO(J)/FO(J,I) ** (RHO(I)-1)));$
$GAMMA(I) = QO(I) / (SUM(J, FO(J,I) * DELTA(J,I) ** RHO(I)) ** (1/RHO(I)));$	$GAMMA(I) = QO(I) / (SUM(J, DELTA(J,I) * FO(J,I) ** RHO(I)) ** (1/RHO(I)));$
$UO=GDPO;$	$UO=GDPO;$
$BETA(I) = CO(I)/GDPO;$	$BETA(I) = CO(I)/GDPO;$
$ALPHA = UO/PROD(I, CO(I) ** BETA(I));$	$ALPHA = UO/PROD(I, CO(I) ** BETA(I));$
$RIO(J,I) = RO(J);$	$RIO(J,I) = RO(J);$
$PIO(J)=1;$	$PIO(J) = (SUM(URBAN, FO(J,URBAN))) / (FBAR(J) - SUM(RURAL, FO(J,RURAL)));$

When we shock the model, we will detail which parameters we change in the text. Assume all unaddressed parameters remain at the levels presented above.

Analysis of Shocks

Simple Increase of the Minimum Wage

The first shock we will consider is an increase of the minimum wage in the city, holding all other parameters at their base case levels.

In the long run Harris-Todaro model presented in Gilbert and Tower, the rental rates of capital in the city and the country must equal one another, because capital is mobile. For example, if the rental rate in the city were higher than the rental rate in the country, capital would simply flow from the country to the city until the two rates equalized. In the long run, then, the urban rental rate of capital must equal the rural rental rate of capital; in the short run, however, because capital is not mobile, it is possible for them to diverge. Furthermore, in both the short and long run, the urban wage will diverge from the rural wage when a distortionary minimum wage is imposed. This is because the urban wage is now fixed at a level above its competitive equilibrium; the market for urban labor cannot clear, which results in a labor surplus, or unemployment. Workers want high-paying city jobs, which are awarded via lottery. If a worker residing in the city does get a job, he will earn the minimum wage; if not, he will be unemployed and earn nothing. Workers therefore flow between the city and the country until the expected value of the urban wage equals the rural wage.

Under our base case assumptions, a small increase in the minimum wage causes production of good 1 to decrease and production of good 2 to increase, which makes sense because factor costs for good 1 are increasing. While we see this production shift in both the short and long run, it is more pronounced in the long run. We ascribe this to the mobility of capital. In the long run, to produce more of the relatively inexpensive good 2, the economy transfers both labor and capital from the city to the country. In the short run, it is constrained to transferring capital; therefore, the shift in production is of lesser magnitude. In keeping with this explanation, we find that urban labor use falls while rural labor use increases in both the short and long run.

One of the most interesting differences between the short run and long run effects of an increase in the minimum wage is that, under our base case assumptions, the rural wage falls while the urban wage rises. We find that, by altering the parameter governing the elasticity of factor substitutability, ρ , we can produce cases in which the rural wage falls in the short run and cases in which it rises. (We examine some of these in more detail below.) However, in the long run, in all the elasticities we consider (with ρ as low as -15 and as high as $.85$), the rural wage increases when a minimum wage is imposed as long as we hold the rest of our parameters at their base case levels. Furthermore, testing a range of elasticities in the short run and long run, we find that the rural wage is consistently higher in the long run than the short run, even when both increase, implying that urban employment consistently falls more in the short run (as the level of the minimum wage is the same and the rural wage is equal to the urban wage times the urban rate of employment). This is because capital cannot adjust; the increase in the minimum wage means good 1 is relatively more expensive to produce, so the economy shifts towards production of good 2 and trades to equalize its consumption. Good 2 is labor-intensive, not capital-intensive, so the (economy-wide) price of capital ends up falling in the long run as a result of this shift. Some capital moves out to the country. However, in the short run, capital can have different prices in the urban and rural sectors, because it cannot flow to equalize the price. Therefore, in the short run, we find that the shift away from production of good 1 causes the urban rental rate of capital (m) to decline precipitously, while the rural rental rate of capital *goes up*. It is in greater demand, after all, and the supply cannot change. This is why capital flows out to the country in the long run, which *increases the rural productivity of labor*, as there is more capital per person, which raises the wage and encourages unemployed workers living in the city to move to the country and take jobs there. *That* is why unemployment increases more in the short run than the long run when the minimum wage increases, holding our other parameters constant.

Very Low and Very High Factor Substitutability

In the short run, when we set the minimum wage at 1.1, leaving all our other parameters at their base case levels, we find that the rural wage declines slightly. The rural wage needs to equal the expected value of the urban wage, which is the probability of acquiring a job (the urban rate of employment) times the minimum wage. If the rural wage falls when the minimum wage is increased, that means that the rate of employment is falling faster than the wage is increasing—the expected urban wage is undergoing a net reduction. This result depends on the level at which we set the substitutability of capital and labor, however. In our base case, the parameter for the elasticity of factor substitution, ρ , is equal to .1, but if it were higher or lower, an increase of the minimum wage would have different effects on employment and therefore on the rural wage. We examine several such cases below.

First, we change ρ from .1 to .8. As σ is equal to $1/(1-\rho)$, this is equivalent to dramatically increasing the substitutability of factors of production. We find that employment falls by slightly more than it does when ρ equals .1. When there is no substitutability between factors of production, capital and labor are only productive in combination; when there is infinite substitutability (i.e. when the factors are perfect substitutes for one another), labor and capital can stand in for one another with no ill effect. Therefore, producers are able to shift labor from the city to the country in greater measure without bringing production of good 1 to a halt. The high degree of substitutability means labor is less essential in production, so the imposition of a minimum wage will cause employment to fall dramatically and thereby decrease the rural wage.


Next, we change ρ from .1 to -15, bringing σ ($1/(1-\rho)$) to a near-zero value, meaning the factors of production are nearly perfect complements (which are productive only in combination). In this case, we find that the rural wage *increases*. In the short run, capital cannot flow between the country and the city, only labor. Because labor and capital are perfect complements, when the use of either factor decreases, the other becomes unproductive. Labor demand is therefore inelastic. When

the minimum wage is increased, producers in the city continue to employ almost the same amount of labor as in the competitive equilibrium. Producers in the country do not want workers to move en masse to the city, rendering their capital unproductive and causing output of good 2 to plummet, so, in a break from our prior experiments, the rural wage increases. Employment is falling by less than the wage is increasing—the expected urban wage is increasing.


An Increase in the Price of the Capital-Intensive Good

We now examine what happens when we pair a binding minimum wage with an increase in the price of good 1 (which is capital-intensive and produced in the city). In both models, we leave the minimum wage at 1.1 and increase the price of good 1 from 1.0 to 1.1, leaving all other parameters, including the price of good 2, at their base case levels. This leads to some extremely interesting effects, particularly for employment. In the short run, we find that, holding the minimum wage at 1.1, employment rises sharply when the price of the good 1 is increased. (Note that this effect is more pronounced when there is greater substitutability between capital and labor. At low levels of substitutability, employment is less affected by the imposition of a minimum wage, so it has less ground to make up.) In the long run, however, an increase in the price of good 1 causes employment to *fall*. We interpret this as follows. In the short run, producers want to make more of good 1 than at the previous equilibrium because it fetches a higher price. They cannot add capital, so the only way they can increase production is by hiring more labor. Labor demand therefore increases, and the minimum wage has less of an impact. Workers flock to the city, where high-paying jobs are suddenly more abundant, and urban labor use rises sharply. However, in the long run, capital migrates to the city—producers are not constrained to increasing production via labor, as they are in the short run, and good 1 is capital-intensive to begin with. Urban capital use therefore rises sharply. The demand for labor falls; while producers still want to produce more of good 1, they no longer rely on the costlier factor of production to do so, especially considering that production of good 1 is capital-intensive. The shift

towards the capital-intensive sector and away from the labor-intensive sector causes unemployment to increase as compared to before the change in the price of good 1.



Short run labor market with sticky wages in the urban sector (sector 1)



Long run labor market with sticky wages in the urban sector (sector 1)

Note: \bar{w}_1 is the wage constraint and the two curves are the labor demand curves for the two sectors; a is the equilibrium with no distortion and b and c are the intersections of the labor demand curve with the horizontal wage level that determine employment.

Appendix A: Our Short Run Model

Below, we present the models we have used in performing our analysis, so that you can examine our equation, replicate our experiments, and conduct your own. First, we have the short run model constructed by Cunnane and Girma.

SET I Goods /1,2/;
 SET J Factors /m,l,n/;
 ALIAS (J, JJ);

SET ST(J) Factors with sticky returns /L/;
 SET NST(J) Factors without sticky returns /m,n/;
 SET RURAL(I) Rural industries /2/;
 SET URBAN(I) Urban industries /1/;

PARAMETERS

ALPHA Shift parameters in utility
 BETA(I) Share parameters in utility
 P(I) Prices
 UO Initial utility level
 CO(I) Initial consumption levels
 XO(I) Initial trade levels
 GAMMA(I) Shift parameters in production
 DELTA(J,I) Share parameters in production
 RHO(I) Elasticity parameters in production
 FBAR(J) Endowments
 QO(I) Initial output levels
 RO(J) Initial factor prices
 FO(J,I) Initial factor use levels
 GDPO Initial gross domestic product
 RIO(J,I) Initial industry level factor prices
 PIO(J) Initial employment rate;

p('1')=1.0;
 P('2')=1.0;
 ro(j)=1;
 QO(I)=100;
 CO(I)=QO(I);
 XO(I)=QO(I)-CO(I);
 FO('L','1')=20;
 FO('L','2')=80;

```

FO('m','1')=80;
FO('n','2')=20;
fbar(J)=SUM(I, FO(J,I));
GDPO=SUM(I,P(I)*QO(I));
rho(i)=.1;
DELTA(J,I)$FO(J,I)=(RO(J)/FO(J,I)**(RHO(I)-1))/(SUM(JJ$FO(JJ,I), RO(JJ)/FO(JJ,I)**(RHO(I)-1)));
GAMMA(I)=QO(I)/(SUM(J$FO(J,I), DELTA(J,I)*FO(J,I)**RHO(I))**(1/RHO(I)));
UO=GDPO;
BETA(I)=CO(I)/GDPO;
ALPHA=UO/PROD(I, CO(I)**BETA(I));
RIO(J,I)=RO(J);
PIO(J)=1;

```

VARIABLES

U	Utility index
X(I)	Trade levels
C(I)	Consumption levels
Q(I)	Output levels
R(J)	Factor prices
F(J,I)	Factor use levels
GDP	Gross domestic product
RI(J,I)	Industry factor prices
PI(J)	Employment rate;

```

U.L=UO;
X.L(I)=XO(I);
C.L(I)=CO(I);
Q.L(I)=QO(I);
R.L(J)=RO(J);
R.LO(J)=0;
F.L(J,I)=FO(J,I);
GDP.L=GDPO;
C.LO(I)=0;
Q.LO(I)=0;
F.LO(J,I)=0;
GDP.LO=0;

```

```

RI.L(J,I)=RO(J);
RI.LO(J,I)=0;
PI.L(J)=PIO(J);
PI.LO(J)=0;
PI.UP(J)=1;
R.FX(ST)=1;
PI.FX(NST)=1;

```

EQUATIONS

UTILITY Utility function
 DEMAND(I) Demand functions
 MAT_BAL(I) Market closure
 PRODUCTION(I) Production functions
 RESOURCE(J) Resource constraint
 LABNST(NST,I) Labor market closure for unregulated factors
 LABSTU(ST,URBAN) Labor market closure for regulated factors in urban
 LABSTR(ST,RURAL) Labor market closure for regulated factors in rural
 FDEMAND(J,I) Factor demand functions
 INCOME Gross domestic product;

```

UTILITY..U=E=ALPHA*PROD(I, C(I)**BETA(I));
DEMAND(I)..C(I)=E=BETA(I)*GDP/P(I);
MAT_BAL(I)..X(I)=E=Q(I)-C(I);
FDEMAND(J,I)$fo(j,i)..RI(J,I)=E=P(I)*Q(I)*SUM(JJ$fo(jj,i), DELTA(JJ,I)*F(JJ,I)**RHO(I))**(-
1)*DELTA(J,I)*F(J,I)**(RHO(I)-1);
INCOME..GDP=E=SUM(I, P(I)*Q(I));
PRODUCTION(I)..Q(I)=E=GAMMA(I)*SUM(j$fo(j,i), DELTA(J,I)*F(J,I)**RHO(I))**(1/RHO(I));
RESOURCE(J)..PI(J)*FBAR(J)=E=SUM(URBAN, F(J,URBAN))+PI(J)*SUM(RURAL, F(J,RURAL));
LABNST(NST,I)$FO(nst,i)..RI(NST,I)=E=R(NST);
LABSTU(ST,URBAN)..RI(ST,URBAN)=E=R(ST);
LABSTR(ST,RURAL)..RI(st,rural)=E=PI(st)*R(ST);

```

```

MODEL SMALL /ALL/;
SOLVE SMALL USING NLP MAXIMIZING U;

```

```

R.FX(ST)=1.1;
SOLVE SMALL USING NLP MAXIMIZING U;

```

Appendix B: Gilbert and Tower's Long Run Model

SET I Goods /1,2/;
 SET J Factors /K,L/;
 ALIAS (J, JJ);

SET ST(J) Factors with sticky returns /L/;
 SET NST(J) Factors without sticky returns /K/;
 SET RURAL(I) Rural industries /2/;
 SET URBAN(I) Urban industries /1/;

PARAMETERS

ALPHA Shift parameters in utility
 BETA(I) Share parameters in utility
 P(I) Prices
 UO Initial utility level
 CO(I) Initial consumption levels
 XO(I) Initial trade levels
 GAMMA(I) Shift parameters in production
 DELTA(J,I) Share parameters in production
 RHO(I) Elasticity parameters in production
 FBAR(J) Endowments
 QO(I) Initial output levels
 RO(J) Initial factor prices
 FO(J,I) Initial factor use levels
 GDPO Initial gross domestic product

RIO(J,I) Initial industry level factor prices
 PIO(J) Initial employment rate;

* Assign values to the parameters

p('1')=1.0;
 P('2')=1.0;
 RO(J)=1;
 QO(I)=100;
 CO(I)=QO(I);
 XO(I)=QO(I)-CO(I);
 FO('L','1')=20;
 FO('L','2')=80;
 *FO('K',I)=(QO(I)*P(I)-FO('L',I)*RO('L'))/RO('K');
 fo('K','1')=80;
 fo('K','2')=20;
 FBAR(J)=SUM(I, FO(J,I));
 GDPO=SUM(I, P(I)*QO(I));
 RHO(I)=.1;
 DELTA(J,I)=(RO(J)/FO(J,I)**(RHO(I)-1))/(SUM(JJ, RO(JJ)/FO(JJ,I)**(RHO(I)-1)));

```

GAMMA(I)=QO(I)/(SUM(J, DELTA(J,I)*FO(J,I)**RHO(I))**(1/RHO(I)));
UO=GDPO;
BETA(I)=CO(I)/GDPO;
ALPHA=UO/PROD(I, CO(I)**BETA(I));
RIO(J,I)=RO(J);
PIO(J)=SUM(URBAN, FO(J,URBAN))/(FBAR(J)-SUM(RURAL, FO(J,RURAL)));

```

VARIABLES

```

U          Utility index
X(I)       Trade levels
C(I)       Consumption levels
Q(I)       Output levels
R(J)       Factor prices
F(J,I)     Factor use levels
GDP        Gross domestic product
RI(J,I)    Industry factor prices
PI(J)      Employment rate;

```

```

U.L=UO;
X.L(I)=XO(I);
C.L(I)=CO(I);
Q.L(I)=QO(I);
R.L(J)=RO(J);
R.LO(J)=0;
F.L(J,I)=FO(J,I);
GDP.L=GDPO;
C.LO(I)=0;
Q.LO(I)=0;
F.LO(J,I)=0;
GDP.LO=0;

```

```

RI.L(J,I)=RO(J);
RI.LO(J,I)=0;
PI.L(J)=PIO(J);
PI.LO(J)=0;
PI.UP(J)=1;

```

```

PI.FX(NST)=1;
R.FX(ST)=1;

```

* Create names for equations

EQUATIONS

```

UTILITY          Utility function
DEMAND(I)        Demand functions
MAT_BAL(I)       Market closure
PRODUCTION(I)    Production functions
RESOURCE(J)      Resource constraint

```

LABNST(NST,I) Labor market closure for unregulated factors
 LABSTU(ST,URBAN) Labor market closure for regulated factors in urban
 LABSTR(ST,RURAL) Labor market closure for regulated factors in rural
 FDEMAND(J,I) Factor demand functions
 INCOME Gross domestic product;

UTILITY..U=E=ALPHA*PROD(I, C(I)**BETA(I));
 DEMAND(I)..C(I)=E=BETA(I)*GDP/P(I);
 MAT_BAL(I)..X(I)=E=Q(I)-C(I);
 PRODUCTION(I)..Q(I)=E=GAMMA(I)*SUM(J, DELTA(J,I)*F(J,I)**RHO(I))**(1/RHO(I));
 FDEMAND(J,I)..RI(J,I)=E=P(I)*Q(I)*SUM(JJ, DELTA(JJ,I)*F(JJ,I)**RHO(I))**(-1)*DELTA(J,I)*F(J,I)**(RHO(I)-1);
 INCOME..GDP=E=SUM(I, P(I)*Q(I));

RESOURCE(J)..FBAR(J)*PI(J)=E=SUM(URBAN, F(J,URBAN))+PI(J)*SUM(RURAL, F(J,RURAL));
 LABNST(NST,I)..RI(NST,I)=E=R(NST);
 LABSTU(ST,URBAN)..RI(ST,URBAN)=E=R(ST);
 LABSTR(ST,RURAL)..RI(ST,RURAL)=E=PI(ST)*R(ST);

MODEL SMALL /ALL/;
 SOLVE SMALL USING NLP MAXIMIZING U;

R.FX(ST)=1.1;
 SOLVE SMALL USING NLP MAXIMIZING U;